
Abstract  
 
Highly accessible HTML files have increased the impor-
tance and use of color as means for enhancing reports 
and summarizing data. The reference manual for 
SAS/GRAPH software contains an entire chapter de-
voted exclusively to color. In fact, the chapter lists 292 
predefined SAS colors by their NAMEs, RGB 
(red/green/blue) and HLS (hue/light/saturation) codes. 
Nevertheless, the list is hampered by the absence the 
colors themselves. VPAPB for “very pale purplish blue" 
continues to remain in the eye of the beholder. 
 
With ODS it is now possible to construct a fully annotated 
color chart from the list of predefined SAS colors found in 
SAS OnlineDoc. Actual colors are displayed in an HTML 
format along with their NAMEs, RGB, and HLS codes. 
 
A color scale new to SAS software and developed from 
the color chart is also described in the paper. The color 
scale uses a decimal conversion macro in a looping struc-
ture to display one hue in varying degrees of lumines-
cence. The paper provides illustrations, instructions for 
construction, and applications of both the color chart and 
the color scale. These visual aids should promote the 
effective use of color for communicating information in 
SAS Software applications. 
 
Constructing a SAS Color Chart 
 
To obtain a listing of the predefined SAS colors appearing 
on pages 146-154 in Version 8 of the SAS/GRAPH man-
ual, download the table from SAS OnlineDoc in HTML 
format into a text file by completing the following steps: 

1) Access the table of contents and drill down to: 
SAS/GRAPH 

SAS/GRAPH COLORS 
Color-naming schemes 

2) Select All and then Copy and Paste to insert the material 
into a Microsoft Word file. 

3) Delete extraneous text and annotations keeping only the 
table. 

4) Select the table, and press Convert Table to Text specify-
ing comma delimited output. 

5) Save the output to a TXT file that is used as input to a 
SAS program. 

The SAS program that processes the downloaded text file 
uses a macro together with traffic lighting to add color 
blocks to output from PROC Report. The macro that is 
invoked from a compute block in PROC Report assigns 
the correct background color to Column #2 with a 
SELECT statement: 
    %macro selectRGB; 
      select (_C2_); 
      %do i = 1 %to &NColors; 
         when (&i.) CALL DEFINE(_COL_,"STYLE", 
           "STYLE=[ BACKGROUND=❶&&RgbN&i.]"); 
      %end; 
      otherwise; 
      end; 
    %mend selectRGB; 

❶ Macro variable &&RGBN&I is created in a Data _NULL_ 
step that processes all 292 colors. It contains an RGB 
code for a particular color. 

Macro variables have to be used for assigning back-
ground colors, because the syntax for the STYLE option in 
the ODS DEFINE statement calls for unquoted constant 
text, not a SAS variable. Here is how the macro is invoked 
in PROC REPORT: 
ods listing close; 
ods html 
  body='c:\SUGI27\SASColorChrt.html'; 
  proc report data=X nowindows headline 
       headskip split='*' ls=100 ps=40; 
    columns cGroup ❶CNum cName cDesc RGB HLS; 
    define  cGroup / order width=15 
                     'Color*Group'; 
    define  cNum   / display width=2  
                     ❷format=missf. 'C'; 
    define cName   / display width=10  
                     'SAS*Color Name'; 
    define cDesc   / display width=30  
                     'Color Description'; 
    define RGB     / display width=8 'RGB'; 
    define HLS     / display width=8 'HLS'; 
    compute CNum; 
      ❸%selectRGB; 
    endcomp; 
    title1 'SAS Color Chart'; 
  run; 
ods html close; 
ods listing; 

❶ The variable CNUM is column #2, the argument of the 
SELECT statement in the SELECTRGB macro. 

❷ CNUM is set to missing with the MISSF format so that 
only the background color is displayed. The MISSF for-
mat is defined with a single range: 

value missf low-high=' '; 
❸ The SELECTRGB macro is invoked from the COMPUTE 

block. 
 
Selected Displays of the Color Chart 
 
A partial listing of uniquely defined SAS colors not belong-
ing to a larger color group is displayed in Figure 1. The 
color description is omitted from the display, since it is 
identical to the SAS color name. Otherwise the SAS code 
used for generating this table is identical to that described 
in the previous section. 
 
Different SAS Color NAMEs can map to the same color 
defined by the RGB code. Figure 2 lists all duplicates by 
name, RGB and HLS codes. The displayed shades con-
firm the presence of synonyms in the predefined color list.  
 
There is also one error in the predefined color list that has 
remained undetected up to now. Technically, the HLS and 
RGB codes for the same color should display identical 
shades. In Figure 3, however, the two color blocks are 
distinctly different for vivid purple. 

Figure 1. A subset of unique predefined SAS colors along with 
their actual shades. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Figure 2. A list of predefined SAS Colors with different NAMES 
that map to the same RGB codes and shades. Note that a single 
RGB code can appear with different HLS codes. Even so, the 
HLS shades are identical to their RGB counterparts in every in-
stance.  
 
  
 
 
 
 
 
 
 
 
 
 
 
Figure 3. HLS and RGB colors should be identical for each SAS 
color name in the predefined list. RGB and HLS codes produce 
different color blocks for vivid purple. 

The Color Scale 
 
The idea for a SAS color scale comes from the out-
standing user-interface supplied by Microsoft PowerPoint 
for adjusting colors. The caption in Figure 4 shows how to 
get to the Custom window in PowerPoint to define a color 
for any object on a slide. The color scale in the window is 
the narrow vertical band to the right of the main color dis-
play. The triangle next to it makes it work like a slider bar. 
By sliding the triangle up or down the color scale, the New 
upper half of the square in the lower right corner becomes 
lighter or darker reflecting the adjacent shade on the bar.  
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Figure 4. To get to the Custom window, press Format:Colors 
and Lines: Color: More Colors: Custom on the menu bar in 
PowerPoint. 
 
When the slider bar is operated in the Custom window, all 
three RGB numbers change. In contrast, only the 
lum(inescence) or light number changes in the HLS scale, 
and the change reflects the position of triangle alongside 
the bar. The value 76 for lum in Figure 4, for example, is 
nearly one third of the distance between 0 and 255. Be-
cause only one component is needed from the HLS code 
to adjust the shade of a given color, it rather than the RGB 
code should be used for deriving a color scale in SAS. 
 
Unfortunately, though, the HLS scale is configured differ-
ently in SAS. Some of the differences can be easily reme-
died. A decimal conversion macro changes a decimal 
number to hexadecimal, and code components can simply 
be rearranged to change a Microsoft HSL into a SAS HLS.  
 
The story is not so simple for the hue component, how-
ever. In Microsoft, values for hue range from 0 to 255 
whereas the maximum value for hue in SAS is 360 reflect-
ing the Tektronix Color Standard. Even when a correction 
is made with interpolation, the same HLS code produces 
different colors in the two software packages. In contrast 
RGB codes are the same, but they cannot be used in this 
instance, because their components change unpredictably 
when a color scale is scanned. 

The Color Scale is derived from the Color Chart 
 
A SAS macro PLTCOLSC is used to create a color scale 
from the list of predefined SAS colors stored locally in a 
permanent SAS data set. The color scale in Figure 5 was 
made possible by matching the macro parameter GOLD 
to its SAS defined HLS code in the data set. Using a fully 
annotated color chart as a guide is the only way to get an 
exact color scale in SAS. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Figure 5. A color scale for GOLD. The vertical line identifies the 
LIGHT component of HLS code for the predefined SAS color. 
  
All hexadecimal numbers in SAS are converted to decimal 
not only for display but because program looping is im-
possible otherwise. Thus for GOLD, H0A080FF in Figure 
1 becomes D160128255 in Figure 5. The structure for the 
HLS code is identified below: 
 

HLS (Hue, Light Saturation) Code 
 Hexadecimal Decimal 
Code H0A080FF D160128255 
Identifying Prefix H D 
Hue (hhh) (ddd) 0A0 160 
Light (LL) (LLL) 80 128 
Saturation (SS) (SSS) FF 255 

 
Numeric conversion macros play an important role in 
working with SAS colors. The programmer works and 
thinks in decimal, but the programs must convert to both 
hexadecimal and decimal when needed. This conversion 
and re-conversion process is apparent in the following 
code fragment from the PLOTCOLSC macro: 
 
  %macro PltColSc(SASCol); 
    proc sql noprint; 
      select hls, cDesc into ❶:hls, :cDesc 
      from rgb.SASColors 
      where cName eq "%upcase(&SASCol)"; 
    quit; 
    %let hlsdec=❷%hlsdec(&hls); 
    %let DHue=%substr(&hlsdec,2,3); 
    %let DLum=%substr(&hlsdec,5,3); 
    %let DSat=%substr(&hlsdec,8,3); 
   %macro scale; 
     data anno; 
      %system(2,2,2); 
      %do i=❸0 %to 255 %by 5; 
         ylow=10; yhigh=20; xleft=&i; 
         xright=xleft+5; 
         %bar(xleft,ylow,xright,yhigh, 
          ❹%hlshex(&DHue,&i,&DSat),0,solid); 

         output; 
      %end; 
       etc. etc. etc. … 
     run; 
   %mend scale; 
   %scale; 
   etc. etc. etc. … 
 %mend PltColSc;  
 %PltColSc(GOLD); 
 
❶ HLS here is a macro variable containing the hexadeci-
mal value for HLS that corresponds to SAS color sent as 
a parameter to the macro. 
❷ HLSDEC is the decimal equivalent for HLS. This value 
is not used directly. Rather, it is parsed into decimal val-
ues for HUE, LIGHT, and SATURATION. 
❸ HUE and SATURATION remain fixed, but LIGHT is 
varied from 0 to 255, minimum and maximum values, in 
increments of 5 (visible as slight ridges in Figure 5). 
Decimal notation is required here so that the looping 
statement will work properly. Increments of 5 are chosen, 
because there is a danger of exceeding the value of 
MAXCOLS for a given graphics device. When the limit is 
exceeded, SAS just supplies default colors in random 
order thereby destroying the color scale that is being cre-
ated. 
❹ The decimal number must be converted back to hexa-
decimal so that SAS can plot the correct color for a given 
increment of the looping variable. 
 
Working with the RGB Code 
 
If only a single color needs to be defined for an applica-
tion, then RGB is the code of choice, since it is configured 
identically in both SAS and Microsoft. In this instance, the 
choice of a color does not need to be confined to a prede-
fined list. Instead, a color can be defined by simply ma-
nipulating the controls in the Custom color window of a 
Microsoft application. Once the color is selected, its RGB 
code must be recorded for processing by the RGBHEX 
macro inside SAS. 
 
The RGBHEX macro converts Microsoft decimal numbers 
to hexadecimal notation that SAS software understands. 
The macro contains only one statement: 
 
%macro RGBHex(rrr,ggg,bbb); 
  %sysfunc(compress(CX%sysfunc(putn(&rrr,hex2.)) 
                    %sysfunc(putn(&ggg,hex2.)) 
                    %sysfunc(putn(&bbb,hex2.)))) 
 %mend RGBHex; 

 
The SAS function COMPRESS is used rather than the 
macro function %COMPRESS to remove all internal 
blanks from the returned hexadecimal color code. For 
example, %RGBHEX(0,204,153) returns CX00CC99CX00CC99CX00CC99CX00CC99. If 
%COMPRESS were used instead, the returned string 
would be unusable as CX00 CC 99CX00 CC 99CX00 CC 99CX00 CC 99. Note that the macro 
statement is missing a semicolon. If one were present, it 
would be returned as a portion of the string constant caus-
ing an error.  
 
Gray is unique 
 
The color gray is automatically produced if the three com-
ponents in the RGB code have the same value. Conse-
quently, the conversion macro GRAYHEX is a simplified 
version of RGBHEX: 

Hue: 160   Saturation: 255

Light
0 20 40 60 80 100 120 140 160 180 200 220 240 260

Color Scale for SAS Color GOLD

%macro GrayHex(gray); 
 %sysfunc(compress(CX%sysfunc(putn(&gray,hex2.)) 
                   %sysfunc(putn(&gray,hex2.)) 
                   %sysfunc(putn(&gray,hex2.)))) 
%mend GrayHex; 

 
Because of this property, the RGB scale can be used to 
construct a color scale for gray. Here is an adjusted ver-
sion of the SCALE macro that works with RGB rather than 
HLS to define the gray scale displayed in Figure 6: 
 
%macro scale; 
  data anno; 
  %system(2,2,2); 
  %do i=0 %to 255 %by 5; 
    ylow=10; yhigh=20; xleft=&i; xright=xleft+5; 
    %bar(xleft,ylow,xright,yhigh,%GrayHex(&i), 
         0,solid); 
    output; 
  %end; 
  %bar(0,10,260,20,black,0,empty); /*outline*/ 
  stop; 
  run; 
%mend scale; 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Figure 6. A gray scale based on the more intuitive decimal nota-
tion enables a developer to define a precise shade for an applica-
tion. 
 
A Table of Conversion Macros 
 
Since the conversion macros are typically brief, they are 
being grouped together in Table 1 below. If the source 
code has already been listed in the paper, a reference is 
made to it. 
 

Name GRAYHEX 
Description Convert an RGB decimal code into Hexa-

decimal format. 

Parms GRAY, a 3-digit zero-filled decimal number. 

Source Code See section Gray is Unique  

Name HLSDEC 
Description Convert a SAS HEX HLS color to its decimal 

equivalent. 

Parms HHHHLLSS a hexadecimal number with a 
leading character H. 

Source Code   
%macro HLSDec(Hhhhllss); 
 %let hue=%substr(&Hhhhllss,2,3); 

  %let ll=%substr(&Hhhhllss,5,2); 
  %let ss=%substr(&Hhhhllss,7); 
  %sysfunc(compress 
      (D%sysfunc(putn( 
         %sysfunc(inputn(&hue,hex3.)),z3.)) 
       %sysfunc(putn( 
         %sysfunc(inputn(&ll,hex2.)),z3.)) 
       %sysfunc(putn( 
         %sysfunc(inputn(&ss,hex2.)),z3.)))) 

%mend HLSDec; 

Name HLSHEX 
Description Convert an HLS decimal code into Hexa-

decimal format. 

Parms HUE: a 3-digit code for hue 
LUM: a 3-digit code for light 
SAT: a 3-digit code for saturation 

Source Code   
%macro HLSHex(hue,lum,sat); 
  %sysfunc(compress 
     (H%sysfunc(putn(&hhh,hex3.)) 
       %sysfunc(putn(&ll,hex2.)) 
       %sysfunc(putn(&ss,hex2.)))) 

%mend HLSHex; 
Name RGBHEX 
Description Convert an RGB decimal code into Hexa-

decimal format. 

Parms RRR: a 3-digit code for Red 
GGG: a 3digit code for Green 
BBB: a 3-digit code for Blue 

Source Code See section Working with the RGB Code  
Table 1. A listing of all the conversion macros used for process-
ing and defining SAS colors. 
 
Applications 
 
•Matching SAS/GRAPH and PowerPoint backgrounds 
The presence of too many colors can lead to “garish, diffi-
cult to interpret and confusing interfaces” according to 
Preece (89). The potential for clashing colors is very great 
when the output from SAS/GRAPH is pasted into a 
PowerPoint Presentation. By matching the PowerPoint 
background color in SAS, it is easy to produce the color-
coordinated tree graph in Figure 7. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Figure 7. Binary tree plots with white selection handles match 
background colors in PowerPoint. The decimal RGB code is cop-
ied from PowerPoint and then matched in SAS by adding 
CBACK="%rgbhex(0,150,136)" to the GOPTIONS statement. 

0 20 40 60 80 100 120 140 160 180 200 220 240 260

Microsoft® PowerPoint Gray Scale in Decimal

•Producing a bar chart where the shades of gray can be 
distinguished from each other 
 
By referencing the gray scale in Figure 6, the bar charts in 
Figure 8 that depict relative speeds of format resolution 
times in graduated shades of gray can be produced with 
the following source code:  
   select (srchtype); 
     when ('S')  color="%GRAYHEX(220)"; 
     when ('SE') color="%GRAYHEX(160)"; 
     when ('BE') color="%GRAYHEX(80)"; 
     when ('B')  color='black'; 
   end; 

With decimal notation, it is easy to remember that low 
numbers are dark and high numbers are light.   
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Figure 8. Relative resolution times for B(inary), B(inary) 
(E)mbedded, (S)equential (E)mbedded, and (S)equential formats 
in ranked order. A discussion of the issues that motivated this 
graph can be found in the paper about the relationship between 
format structure and efficiency cited in the reference section. 
 
•Producing a visually appealing, more informative contour 
plot 
 
Preece's quote about too many colors is particularly ap-
propriate when SAS default colors are used for generating 
contour plots. Two versions of the same contour plot are 
displayed in Figures 9 and 10 at the end of the paper. The 
source code and data come from the SAS Sample Li-
brary. 
 
A simple loop in the source code is used to plot the con-
tours in varying shades of blue and brown. Since the col-
ors are related to each other, the HLS scale must be used 
with a color obtained from the annotated list of predefined 
SAS colors. This particular version of the list also lists 
HLS codes in decimal. Again the decimal number is con-
verted to hexadecimal inside the loop so that the correct 
SAS color is assigned. The relevant code fragment ap-
pears below: 
 
    /* color the contour plot */ 
  %macro c19; 

   %let j=1; 
   %do i=80 %to 190 %by 10; *water; 
      pattern&j value=msolid  
        color=%hlshex(001,&i,145) ; 
      %let j=%eval(&j+1); 
   %end; 
   %do i=180 %to 60 %by -20; *land; 
      pattern&j value=msolid 
        color=%hlshex(152,&i,045) ; 
      %let j=%eval(&j+1); 
   %end;      
 %mend c19; 
 

Summary and Conclusions 
 
Two visual aids, the color chart and color scale, have 
been presented in this paper. The color chart is used to 
enhance the SAS list of predefined colors by associating 
an actual color with its NAME, HLS and RGB code. The 
color scale developed from the color chart displays one 
hue in varying degrees of luminescence. The color scale 
is particularly helpful for designing graphs that depict 
magnitude through a meaningful arrangement of related 
colors. Finally, here is the actual color associated with 
VPAPB - hardcopy in a shade of gray. 

Figure 9. First rendition of Mill Pond is garish in full color. There is no rationale behind coloring the 
deepest part of the pond red, and the shallowest part yellow. Some study is also required to realize 
that dashed lines are used for depicting dry land. 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Figure 10. In the revised contour plot, water depth is depicted in varying shades of blue and land 
height takes on darker shades of brown at higher elevations. There is no need for a legend, since all 
communication is conveyed in the title. 

